
Everyone, it seems, knows how to play Noughts And Crosses.  This project sets out to implement this game on the computer.  It extends the ideas developed in the previous two projects, particularly the way you can write a program to check a situation (two cards, a noughts and crosses board) to see whether certain conditions (a pair of cards, three of the same symbol in a row) have been met.  These sorts of procedures find serious use in many different applications.



This Project also introduces the concept of a function in Visual Basic.  Functions are like the general procedures you have been using - they are defined in the general section of the code window, and are available to any procedure in the program.  Functions, however, return a value.  You call a function in code, and get back some value (a number, one of true or false, a string of text) which you can then use in your program.  You should be familiar with mathematical functions already; Visual Basic functions are more general, and you can define them for yourself.









	1.	Investigation





Noughts And Crosses is a simple game that most people know about.  It is a good method of developing systematic thinking and logical reasoning in young children.  A computer Noughts and Crosses game would prove very popular.





	Problem Statement:	 To implement a two-player Noughts and Crosses game on the computer.  The game should allow players to click on a cell of the Noughts and Crosses grid, and have their symbol appear in that cell.  The program should also be able to detect when one player has won the game, and provide a suitable message.





We will keep the interface simple at this stage,  not worrying too much about things like the players’ names, or very sophisticated messages to the players.  The first player will always play noughts, and the players will not be able to choose their symbols.

	2.	Design





The User Interface





A Noughts and Crosses grid is fairly easy to draw on a Form using the Line control tool.  The players need to be able to click on the cells to enter their symbol, so the Form needs nine controls (one in each cell) that can respond to the Click event.  The example below uses Label controls, so that the players can’t enter text into them directly.



The Form might look like this (one of the Labels has been selected to show its position on the Form):



��		Label

In play, the Form should look like this:

Program Operation



There is only one action which a user can perform on this Form - clicking one of the Label controls.  The winner will be announced in a Message Box.



	Event:	The program starts



	Actions:	1.	Constants are declared to be used to represent the state of a cell (“nought”, “cross”, or “empty”).



		2.	Variables are declared to store the current player’s symbol, the opponent’s symbol, and the current state of the grid.



		3.	The variables are initialised, and the grid blanked.





	Event:	The user clicks one of the label controls



	Actions:	1.	The program checks to see if the cell is empty.  If it is not, a beep sounds.



		2.	If the cell is empty, the player’s symbol is placed in the cell, and the player and opponent variables have their values swapped.



		3.	The program checks to see if the player who has just moved has won the game, and displays a suitable message if she has.









	3.	Production





Creating The Form And Its Controls





The Line controls have their BorderWidth property set to 4.



The Image controls are in a control array, named something like cell.  Their BackStyle property needs to be set to 0 - Transparent, so that they do not cover the Lines. 



You will need to set the FontSize property to its largest possible value.  Experiment with different fonts, sizes and alignments to get a symbol which is large enough, and somewhere close to the centre of the cell.







Entering The Program Code





The program will need a general procedure to check whether a player has won.  This will be a function; the program will give the function a value to represent the current player, and the function will return one of the values True or False, to indicate whether that player has won the game or not.



There will also be a general procedure to update the grid each time a player has a turn.



�Const nought = "0", cross = "X", empty = ""

Dim player As String

Dim opponent As String

Dim grid(0 To 8) As String



Sub Form_Load ()

    player = nought

    opponent = cross

    For counter = 0 To 8

        grid(counter) = empty

    Next counter

    Form1.Caption = player + " to play"

End Sub



Sub cell_Click (index As Integer)

    If grid(index) = empty Then

        grid(index) = player

        updategrid

        If CheckForWin(player) = True Then

             message = "Congratulations"

             message = message + Chr$(13) + Chr$(10) + "Click on OK to exit"

             MsgBox message, 64, "YOU WON!!"

              End

         End If

         temp = player

         player = opponent

         opponent = temp

         Form1.Caption = player + " to play"

    Else

        Beep

    End If

End Sub



Sub updategrid ()

    For counter = 0 To 8

        cell(counter).Caption = grid(counter)

    Next counter

End Sub



Function CheckForWin (who)

    win = False

   For counter = 0 To 2

   If (grid(3 * counter) = who) And (grid(3 * counter + 1) = who) And (grid(3 * counter + 2) = who) Then

            win = True

    End If

    Next counter

    For counter = 0 To 2

        If (grid(counter) = who) And (grid(counter + 3) = who) And (grid(counter + 6) = who) Then

            win = True

        End If

    Next counter

    If (grid(0) = who) And (grid(4) = who) And (grid(8) = who) Then

        win = True

    End If

    If (grid(2) = who) And (grid(4) = who) And (grid(6) = who) Then

        win = True

    End If

    CheckForWin = win

End Function





	4.	Evaluation





Testing The Program





Play several games by yourself, and check that it is operating as you expect it to.  In particular, consider the following aspects of the game:



1.	Does the program allow you to enter a symbol into any empty cell?  



	Does it allow you to enter a symbol into a cell that already has a symbol in it?





2.	Does the program always detect a win?  Check for all of the possible winning combinations, for both players.





Improving The Program





1.	Make the program a little more friendly.  Have it ask for the players’ names at the start of the program, and adjust the Form caption and the winner message box to refer to the players by name.





2.	Have the program allow the players to choose their symbols at the start of a game.





3.	Have the program ask whether the players want to play another game.  You can have different players, or the same players again.



	If your program allows the same players to play again, arrange the program to keep score of how many games each player has won.

Extension Project:    “The Computer Plays Noughts And Crosses”





The theme of this Extension is to get the computer to take the part of one of the players in the Noughts and Crosses game.  That is, you are being asked to teach the computer how to play the game.





You will need to provide a way of asking the user if the game is between two humans, or a human and the computer. You will also need to have some way of determining which player - human or computer - moves first, and which symbol each will use. If you have programmed the Improvements to the Noughts and Crosses game, these modifications should not present a problem.





The difficulty arises when you write the code for the computer’s moves - how does the program determine which cell to use?  The program already has an array (the grid array) storing the state of the grid.  From this array, your program can identify the empty cells, and thereby produce a list of possible moves.  But which move to choose?  You will need to have the program able to block an the opponent from winning, as well as identify the cells which are most likely to lead to a winning situation for the computer.



You will need to be systematic.  The program can be written to test for all possible situations, but the code would be very long, and enormously complicated.  You will need to know, yourself, which are the best moves to make, and have the computer look for them first.





Some things to consider when devising a strategy:



	How many winning combinations are there?  How many times does each cell feature in a winning combination?



	How many moves are there in a game of Noughts and Crosses?



	There are only three different types of cell on the grid: the middle cell, the corner cells, and the middle of the sides.  How many times does each type feature in a winning combination?



	What is the best first move?



	If the opponent moves first, what is the best second move for each of the possible first moves?



	How many different combinations of first and second move are there?  What is the best third move in each case?











If you want a real challenge, try some Artificial Intelligence programming.  Write a program where the computer is one of the players in a series of games of Noughts and Crosses.  When the program is first run, have the program choose from amongst the available empty cells at random.  Have it record its moves for each game.  At the end of a game, have the program change its selection process by altering the probabilities of choosing the different cells.  If the computer has won the game, have the choices it made for that game increase their probability.  If the computer loses, have the probabilities decrease.  If the game is a draw, leave the probabilities unchanged



Can you get the computer to learn the strategy for playing Noughts and Crosses?



page � PAGE �112�	Programming In Visual Basic



	Noughts And Crosses	page � PAGE �111�





Noughts And Crosses



Chapter

14








