
Drawing and graphics programs are amongst the commonest and most popular software packages for personal computers.  Most personal computers come with a drawing program already installed (the Microsoft ‘Paintbox’ program is included with the Windows operating environment, for instance), and most computer users have at least played with a drawing package.  In this project you will create a very simple drawing package for yourself.



The Project introduces several new and important ideas.  All of the Projects which you have created so far have only used one Form.  A Visual Basic Project can contain many Forms, and this Project uses one Form for the drawing part of the program, and another Form is used to save the images created by the drawing program.  The Project also makes use of this second Form to load images previously saved by the program.  Sharing data between several Forms requires the use of a Visual Basic module, which is a type of Visual Basic file which you have not yet used.



The Projects you have written to this stage in this course have only responded to a limited range of events: the user clicking on a control (Command Button, Label) on the Form; or the user clicking on a Menu item.  The only other input has been text which the user has typed into a text box.  This Project involves responding to mouse events - pressing a mouse button down, holding the mouse button down, moving the mouse, and releasing the mouse button.  These events add a dynamic aspect to your application.



This Project also involves the use of the graphics methods available in Visual Basic.  You have used some of the graphics objects before, but so far you have created the objects at design time, and your program has only changed the properties of the objects.  In this Project, the objects are created as the program is running, again making for a more dynamic application.



Lastly, and perhaps most importantly, this Project involves saving on disk (and retrieving from disk) the data created by the program.  Using data files is a significant step forward in your programming knowledge.  With this ability, you increase the range of tasks which your applications can perform, and you add a whole new dimension of usefulness to those applications.





The Project is, by necessity, very simple and unsophisticated in the graphics tasks which it can perform.  You will, however, be surprised at the ease with which you can create a program to produce graphic effects.





�

	1.	Investigation





Visual Basic has an extensive set of graphics controls available for you to use in your applications.  There are also many graphics methods which you can use in your programs.  These graphics tools can be combined in a useful and interesting way in a drawing application.





	Problem Statement:	To implement a simple drawing program in Visual Basic.  The program should allow the user to draw lines of different thicknesses on a Form, and to erase these lines.  The user will also be able to save the image in a data file on disk, and retrieve such data files from disk at a later stage.





The application should make use of menus for the user to specify the attributes of the drawing, and to activate file saving and retrieving.  This will leave the screen clear for use as a drawing area.









	2.	Design



The User Interface



The first Form used in this application is the main drawing Form.  There are no controls on the Form, just a Menu Bar:

The menus contain the following options:

The second Form used in this application is used for file input and output.  It contains several controls which you have not yet met:



������		Labels			Text Boxes		Command Buttons





�������





	File List Box	Combo Box	Labels		Drive List Box	Directory List Box





The File List Box, Drive List Box, and Directory List Box are controls for accessing the drives, directories, and files  available on your system.  This Form has been designed to look as much like a Windows File management dialog box as possible.  This helps your users program by maintaining the uniformity of the Windows interface.





Program Operation



Most of the actions in this application, apart from the menu selections, are to do with mouse movements and mouse button operations.





	Event:	The program starts



	Actions:	1.	A general variable, to register whether the program is in ‘drawing’ mode or not, is declared.



		2.	The mouse pointer is set to the cross-hair shape.





	Event:	The user presses the mouse button down



	Actions:	1.	Sets the drawing mode variable to true.  Pressing the mouse button causes drawing to occur.



		2.	Store the horizontal and vertical co-ordinates of the mouse pointer.  The position of the mouse when the button is pressed becomes the starting point of a new line. 



	Event:	The user moves the mouse



	Action:	1.	If the program is in drawing mode, then draw a line from the last position to the current position of the mouse.





	Event:	The user releases the mouse button



	Action:	1.	The drawing mode variable is set to false.  Releasing the mouse button indicates that drawing is to cease.





	Event:	The user chooses New from the File menu



	Action:	1.	The current picture is cleared





	Event:	The user chooses Open from the File menu



	Action:	1.	Display the Open File dialog box.





	Event:	The user chooses Save from the File menu



	Action:	1.	Display the Save File dialog box.





	Event:	The user chooses the Pencil from the Tools menu



	Actions:	1.	Set the computer to draw in black.



		2.	Set the mouse pointer to the cross hairs.





	Event:	The user chooses the Eraser from the Tools menu



	Actions:	1.	Set the computer to draw in white.



		2.	Set the mouse pointer to a box shape.



		3.	Set the drawing width to a large value (to make erasing easier).





	Event:	The user chooses a width from the Thickness menu



	Action:	1.	Set the drawing width to an appropriate value.





There are a lot of procedures here, but each one is very short.  Most of them, in fact, will consist of only one command.

	3.	Production





Creating The Form And Its Controls





1.	The first Form is easy enough to create.  Give it a suitable name (such as Pad).  



Set its AutoRedraw property to True.  This ensures that if anything (like the file Save and Open dialog boxes) is placed in front of the Form, the Form (and its graphic image) will be redrawn when the dialog box disappears.



Set its MousePointer property to 2 - Cross.



Set its WindowState property to 2 - Maximized.  This will cause the Form to be automatically maximised when the program is run.





2.	Create the Menu Bar.  Give the menu items suitably descriptive names.





3.	To create the second Form, choose New Form from the Visual Basic File menu.



	Name the Form something like FileIO (for File Input/Output).





4.	The controls for this Form are easy enough to create.  



	The Drive List Box, Directory List Box and File List Box do not need to have any of their properties reset from their default values.  The actual data which appears in them will depend on the settings for your particular system at the time you create the controls.



	The Combo Box has its Style property set to 2 - Dropdown List.  The text which appears in this box at design time is the Name of the control. 





5.	Save the Project, naming the Forms something like PENCLBOX.FRM and FILEIO.FRM.  You can name the Project file something like PENCLBOX.MAK









Entering The Program Code





All of the programs which you have written up to now have had a single Form.  The code for those programs has been attached to that one Form.  Some of the code (the general declarations and general procedures) have been attached to the Form itself, and available across all of the event procedures for the Form.  Other code (the event procedures) has been attached to particular controls on the Form, and has been available only to that event.



This Project has two Forms, and there will be code attached to each Form.  In addition, there will be declarations (and perhaps code) which needs to be available to both of the Forms.  Code which is available to all of the Forms in a Project needs to be placed in a project Module.

1.	Choose New Module from the Visual Basic File menu.  You will see a Code Window named Module1.bas:

Into this window, enter the code for the module:



Global filename As String

Global filemode As String



This code declares two global variables - variables whose values are available to any Form or procedure in the entire Project.  The filename variable stores the name of the data file containing the image being saved or loaded.  The filemode variable indicates whether the user has chosen to Save or Load a data file; this variable is necessary because the same Form is used for both operations.



Save the Project again.  You will be asked if you want to save changes to Module1.bas.  Answer yes, and name the file something like Pencil.





2.	Switch to the Project window.  You should see the names of the three files which make up this Project:

	You need to take care to attach the rest of the code to its appropriate Form.





3.	Switch to your PENCLBOX Form - the main drawing Form with the menus on it.



	Add this code to the general declarations section of the Code Window:



Dim drawing As Integer



	This declares a variable which stores whether or not the program is in ‘drawing’ mode.



	The next four procedures all belong to the Form, but for different events on that form:



Sub Form_Load ()

    pad.MousePointer = 2

End Sub

Sub Form_MouseDown (Button As Integer, Shift As Integer, X As Single, Y As Single)

    drawing = True

    currentx = X

    currenty = Y

End Sub



Sub Form_MouseMove (Button As Integer, Shift As Integer, X As Single, Y As Single)

    If drawing Then

        Line (currentx, currenty)-(X, Y)

    End If

End Sub



Sub Form_MouseUp (Button As Integer, Shift As Integer, X As Single, Y As Single)

    drawing = False

End Sub



These last three procedures are interesting - the event in each case passes values to the procedure.  These values indicate which mouse button was clicked, whether the Shift key was depressed, and the X and Y co-ordinates of the mouse when the event occurred.



Now for the menu code:



Sub MnuNewFile_Click ()

    pad.Cls                                          'Clear the screen of any graphics objects

    pad.Picture = LoadPicture()           ‘Clear the screen of any pictures 

End Sub





Sub MnuOpenFile_Click ()

    filemode = "Loading"

�    fileio.Caption = "Open File"

    fileio.Show

    fileio.FileBox.Enabled = True

End Sub



Sub MnuSaveFile_Click ()

    filemode = "Saving"

    fileio.Caption = "Save File"

    fileio.Show

    fileio.FileBox.Enabled = False

    fileio.FileNameBox.SetFocus

End Sub



Sub MnuExit_Click ()

    End

End Sub



Sub MnuPencil_Click ()

    pad.DrawMode = 1

    pad.MousePointer = 2

End Sub



Sub MnuEraser_Click ()

    pad.DrawMode = 16       ' Draw in white

    pad.MousePointer = 4    ' Cursor becomes a rectangular shape

    pad.DrawWidth = 8       ' Draws a fairly wide line

End Sub

Sub MnuThick1_Click ()

    pad.DrawWidth = 1

End Sub



Sub MnuThick2_Click ()

    pad.DrawWidth = 2

End Sub



Sub MnuThick4_Click ()

     pad.DrawWidth = 4

End Sub



Sub MnuThick8_Click ()

     pad.DrawWidth = 8

End Sub





4.	Now use the Project window to switch to the FILEIO form.  Enter this code:



Sub Form_Load ()

    typebox.AddItem "Bitmaps (*.bmp)"               These two lines set up the list of file types in 

    typebox.AddItem "All files (*.*)"                       the combo box at the bottom of the Form.

    typebox.ListIndex = 0                                        Sets the list’s first item as the one to show.

    DirDisplayBox.Text = DirectoryBox.Path

    FileBox.Pattern = "*.bmp"                                 Sets the type of file to show in the File List.

    cancelbutton.Default = True

End Sub



Sub CancelButton_Click ()

    FileName = ""

    fileio.Hide

End Sub



Sub OKButton_Click ()

    FileName = FileNameBox.Text

    If FileName <> "" Then

        If filemode = "Saving" Then

            SavePicture pad.Image, DirectoryBox.Path + "\" + FileName

        Else

            pad.Picture = LoadPicture(DirectoryBox.Path + "\" + FileName)

        End If

    End If

    fileio.Hide

End Sub



Sub DirectoryBox_Change ()

    FileBox.Path = DirectoryBox.Path

    DirDisplayBox.Text = DirectoryBox.Path

End Sub



Sub DriveBox_Change ()

    DirectoryBox.Path = DriveBox.Drive

    DirDisplayBox.Text = DirectoryBox.Path

End Sub



Sub FileBox_Click ()

    FileNameBox.Text = FileBox.FileName

End Sub



Sub TypeBox_Click ()

    If typebox.Text = "Bitmaps (*.bmp)" Then

        FileBox.Pattern = "*.bmp"

    ElseIf typebox.Text = "All files (*.*)" Then

        FileBox.Pattern = "*.*"

    End If

End Sub





5.	Save the project.







	4.	Evaluation





Testing The Program





1.	Run the program.  Notice the appearance of the Form on the screen, and the appearance of the mouse cursor.



	Hold the mouse button down, and move the mouse.  Release the mouse button, and again move the mouse.





2.	Select the eraser Tool.   Hold the mouse button down as you move the mouse over your drawing.





3.	Select New from your File menu.



	Now select a different item from the Thickness menu, and use the mouse to draw on the screen.



	Try all of the different sizes.





4.	Check that the Exit option works on your File menu.





Does the program behave as you expected it to behave?



Does it behave like other drawing programs you have used?





5.	Run the program again, and draw an image on the screen.



	Select Save Picture... from your File menu.  



	Experiment with the different controls in your Save File window.



	Click the Cancel button.





6.	Select Save Picture... again.  Set a suitable drive and directory, and give your file a suitable name (such as TEST1.BMP).  Click the OK button.



	Select New Picture from your File menu to clear the screen.



	Now choose Open Picture... from your File menu.



	Set the same path as you used to save the file.  Make sure that the file name appears in the File List Box.



	Select this file, and click the OK button.





Does the File Management window work as you expected it to work?





7.	Try making some errors in the use of the File Mangement window, such as using an invalid file name when saving a file.  Notice what happens to your program.







Improving The Program





Your testing should have suggested several improvements which could be made to your program.  It should also have indicated that your program needs some “error trapping” built into the File Management code befor the program could safely be given to an inexperienced user.





1.	Add a variable to the program which registers whether any changes have taken place on the current drawing.  Use this variable to warn your user whenever an action is chosen (such as Opening a file or starting a New picture) which would result in the loss of the picture.  Have the program allow the user the choice of saving the picture before continuing. 





2.	Add a Colour menu to the program, so that the user can set the colour of the line.





3.	Add a menu to allow the user to draw straight lines, rectangles, or circles.

Extension Project:	“Expanded Pencil Box”





The last suggestion for Improving the Pencil Box program is in fact quite difficult.  The theme of this extension is to expand the capabilities of the Pencil Box program to include the capacity to draw Lines, Circles and Ellipses, and Rectangles and Squares.  The program should also allow the user to set the Fill and Border colours for a shape, and to set a Fill Pattern.





	Problem Statement:	To create a drawing program with the capacity to produce different geometric shapes, such as circles, squares, and straight lines, as well as freehand lines.  The program should also allow the user to set different colours and patterns for the objects produced by the program.











Take these enhancements one at a time.  The most difficult programming will come in having the program  create the other Draw objects.  Check the properties of these objects carefully, and think about how to use the co-ordinates generated by the different mouse events.  In particular, think about how to use the Top, Left, Width and Height properties of the objects.





You might like to experiment with having the program include some of the icons available in the Visual Basic software.  The user selects an image from a menu, and an icon with that image is then placed wherever the user clicks the mouse button.





You should retain the file saving and loading capabilities of th Pencil Box program.  Read the Visual Basic Help information about errors and error trapping to include some features in your program which prevent it ‘crashing’ whenever an error occurs.  At least, have your program detect a file name which is too long, or one which does not have the proper .BMP suffix.  You could try and have the program add the suffix for the user.

“The Pencil Box”



Notice how these two procedures set properties and events on the other Form.

This is done by placing the name of the Form (fileio, in this case) in front of the name of the control or property.

The Show command makes the other Form visible.








