
This Project involves another game.  This time, it is the ‘memory’ card game.  There are many variations on this game, and it goes by many names - “Concentration”, “Pairs”, “Memory” and so on.  The game can be played with an ordinary pack of playing cards, or with special decks of cards.  It is often used in the education of very young children as a pleasant way of reinforcing letter and number recognition, and of developing classification skills.



Basically, the game is played with a pack of cards consisting of sets (2 or 4 cards) of identical cards.  The cards are laid out face down, and the players take turns at turning over two cards.  If the cards are a pair, the player takes them and has another turn.  If they are not, the are turned face down again, and the other player has a turn.  The winner is the player who collects the most pairs.









	1.	Investigation



The Memory Game is extremely popular with young children, and can be easily implemented on the computer.  At this stage, the game will be set up with pairs of icons from the library supplied in the Visual Basic software.





	Problem Statement:	To implement the Memory Game, using graphic icons to represent the card ‘faces’.  The game will be simplified into a one-person game, with the player selecting pairs until no more remain.  The player’s score will be the number of turns required to collect all of the pairs.





Programming the simplified form of the game will take care of all of the major programming problems.  Keeping score, and developing the two-person game, will be left until later.









	2.	Design





The User Interface





The Form requires a collection of Image controls containing graphic icons.  These images are then covered by a collection of buttons, thereby hiding the icons.  There are no other controls needed, as the game operates with the player clicking on the buttons to reveal the icons.

A suitable Form might contain these Image controls:





�





covered by buttons:

Program Operation



There is only one action the a user can perform on this Form - clicking one of the buttons.  The user will be offered the opportunity to play the game again after all of the pairs of ‘cards’ have been exposed.





	Event:	The program starts



	Actions:	1.	Variables are declared to store the number of pairs of ‘cards’, whether or not there is already a ‘card’ showing, and the index numbers of the two ‘cards’ clicked.



		2.	Set the number of pairs (twelve in the example above).



		3.	Make sure that all of the ‘cards’ are visible.



		4.	Register that there is not a ‘card’ showing







�	Event:	The user clicks a ‘card’



	Actions:	1.	Show the picture under the ‘card’.



		2.	Check to see whether there is already a ‘card’ showing.



		3.	If a ‘card’ is already showing, store its index number as the second card, register that there is not a ‘card’ showing, and check to see if the two ‘cards’ match.



		4.	If the two ‘cards’ do match, remove them from the form, and reduce the number of pairs by 1.  If the ‘cards’ don’t match, hide the pictures.



		4.	If a ‘card’ is not showing, register that there is now a ‘card’ showing, and store its index number as the first ‘card’.



		5.	If all pairs have been uncovered, ask if the player wants to play again.





The player will be asked if she wants to play again in a Message Box.  The Message Box will have a Yes button and a No button - clicking the Yes button will run the initialisation procedure, while clicking No will end the program.







	3.	Production





Creating The Form And Its Controls





When creating the Form, make sure to place the Image controls into a control array.  Name the array something like face (since the images represent the faces of the ‘cards’).  The icons used in the example above come from the C:\vb\icons directory, and from the elements, industry and misc subdirectories.



The program is going to need some way to check whether two ‘cards’ have the same picture on them.  You can use the Tag property of the Image controls to do this.  The Tag property can be anything the programmer wants it to be, without affecting any of the other properties.  Each pair of Image controls with the same picture should be given the same Tag property value (something like umbrella, or earth, or rocket).  Make sure that each Image in a pair has exactly the same Tag.



The Command Buttons should also be in a control array, named something like card.  







Entering The Program Code





Because the buttons are in a control array, the same Click procedure applies to all of them.  This is one of the most important reasons for putting controls in arrays - it makes the programming much simpler.  Whenever any one of the buttons in the array is clicked, the same event procedure is run.  Visual Basic registers the index number of the particular button which was clicked, and makes it available (in a variable) for the procedure to use.



The Card_Click procedure will be kept as simple as possible by using general procedures to check for a match,, and to check whether all of the pairs have been exposed.





Dim NumberOfPairs As Integer

Dim showing As Integer      '  True or False

Dim firstcard As Integer

Dim secondcard As Integer



Sub Form_Load ()

    initialise

End Sub



Sub initialise ()

�    NumberOfPairs = 12

    For counter = 0 To (NumberOfPairs * 2) - 1

        card(counter).Visible = True

    Next counter

    showing = False

End Sub



�Sub card_Click (index As Integer)

    If showing = False Then

        card(index).Visible = False

        firstcard = index

        showing = True

    Else

        card(index).Visible = False

        secondcard = index

        showing = False

        CheckForMatch firstcard, secondcard

    End If

End Sub



�Sub CheckForMatch (fst As Integer, snd As Integer)

    If face(fst).Tag = face(snd).Tag Then

        NumberOfPairs = NumberOfPairs - 1

        face(fst).Visible = False

        face(snd).Visible = False

    Else

        card(fst).Visible = True

        card(snd).Visible = True

    End If

    CheckForQuit

End Sub



�Sub CheckForQuit ()

    If NumberOfPairs = 0 Then

      Answer = MsgBox("Do you want another game?", 4, "Game Over")

      If Answer = 6 Then

           initialise

      Else

           End

      End If

    End If

End Sub









�	4.	Evaluation





Testing The Program





1.	Play the game several times.  Make sure that the cards are disappearing when the pictures match, and that the buttons reappear to cover the pictures when the pictures do not match.





2.	The game is boring, isn’t it?  The pictures are always in the same place each time, and there is no challenge at all in playing the game.  It will be a much better game if the ‘cards’ are shuffled at the start of each game, so that they appear in different places each time the game is played.



	Creating a shuffling algorithm is one of the Extensions for this Project.









Improving The Program





1.	You may have found that the pictures did not stay on the screen long enough to see what they were.  They either disappeared completely (presumably, they matched), or the button reappeared.



	Add a ‘delay loop’ to the CheckForMatch procedure, to slow things down:



Sub CheckForMatch (fst As Integer, snd As Integer)

     For counter = 1 to 10000

     Next counter

    If face(fst).Tag = face(snd).Tag Then

        NumberOfPairs = NumberOfPairs - 1



	This loop simply causes the computer to count to 10,000 before executing the rest of the procedure.  Experiment with the number in the loop until you get an acceptable delay.





2.	Add a box to display feedback to the user.  Have a suitable message appear in this box when a pair is found.





3.	Have the program record the number of attempts taken by the player, and display the total number in the feedback box at the end of the game.















�Extension Project:	“Two-Player Memory Game”





The task for this Extension Project is to develop the Memory Game so that two people can play it.  This method of play will not be very interesting unless the ‘cards’ are shuffled before each game, so the first task is to develop a shuffling procedure.  This procedure is then called from the initialise procedure:



Sub initialise ()

    NumberOfPairs = 12

    For counter = 0 To (NumberOfPairs * 2) - 1

        card(counter).Visible = True

    Next counter

    showing = False

    shuffle

End Sub





	Problem Statement:	To create a procedure which will place a set of ‘cards’ into a random order, and to incorporate this into the Memory game.  Also, to modify the Memory Game so that it can be played by two players.  The two-player memory Game will keep score for each player, and display each player’s score on the screen.





A Form for this Game might look like this, part-way through a game:





�





The rules for the two-player game allow a player to have another turn if a pair has been successfully exposed.  The program will have to keep track of which player is taking the turn, and adjust its scoring accordingly.





The difficult part of the program is the shuffling algorithm.  There are several well-known algorithms for performing this task.  To help you develop your own algorithm, try carrying out the process yourself in some physical manner - you might use pieces of paper to represent the cards, and write the name of the picture on the paper.  Alternatively, you can just draw boxes on a piece of paper, and write numbers into the boxes.



A useful starting point:  work through the boxes (or cards) one by one.  For each one, swap its contents with another box, chosen at random.

Note the advantage of the control array - the ‘cards’ can be made visible very easily using the For ... Next  loop.



The variable named index stores the index property of the particular element of the card control array that was clicked.



firstcard and secondcard store the index numbers of the two cards.  These numbers are then passed to the CheckForMatch procedure.



When the CheckForMatch procedure is defined, it is told to expect two integers to be passed to it.  It calls these integers fst and snd for the purposes of writing the procedure.



The number 4 in the MsgBox function places two buttons in the box.  The value stored in Answer indicates which of these buttons was clicked.

A value of 6 indicates the yes button.








