
Most of the computer capacity in the world today is devoted, in one way or another, to the management of databases.  A database is a collection of data organised into records, with each record consisting of several fields.  The computer is used to manage this database - add and delete records, update data, search for data, select certain data, produce printed output from the database, and so on.  Because of the large number of computerised databases in use, database applications are an important area of software development.



This Project gets you to construct a simple database program.  Obviously, the program will not be particularly sophisticated, but it will introduce you to the main concepts of database software.  



The project requires the use of a user-defined data type.  Up until now, all of the declarations of variables have included a specification of the data type (string, integer, etc), and these data types have been the built-in data types recognised by Visual Basic.  For a database, you need to define your record structure as a data type, and this type will be unique to your particular database.



The Project also involves the use of an external data file.  The “Pencil Box” program used an external data file, but that file stored a single graphic image.  You saved an image in the file, or you loaded an image from a file; in each case, you dealt with the file once, and dealt with the complete file all at once.  This database program, on the other hand, requires interaction with the data file - updating the data every time a change is made to the database, for instance.  In this application, the database itself is never completely loaded into the computer’s memory.  The program only acts as a sort of “viewer” into the data; it only displays a single record at a time.



The procedures used in the database also differ significantly from other procedures you have written.  The data in the database is stored in alphabetical order, and new records have to be inserted in the correct place in that order.  This process requires searching the database, and comparing items of data.  These techniques are also required when the program is trying to find particular items of data, and when a database is being sorted into some different order.



Many of the procedures make use of string functions in Visual Basic - functions which allow you to write code which manipulates text data.











�

	1.	Investigation



Databases come in many different varieties, but they all have the common features of records and fields.  This database application stores a “birthday book” - a collection of names and birthdates.  There is one record for each person in the database.  Each record contains the person’s name, their data of birth, and a comment.  The database is saved in a disk file which is opened each time the program is used, and saved each time a change is made to the database.



	Problem Statement:	To create a simple “Birthday Book” database.  The application will store names, dates of birth, and a comment for a collection of people.  The program will allow new records to be added to the database, and will find a record in the database. It will also allow the user to move systematically through the data.  The data base will be stored in a disk file which the program will open and save as necessary.



The program will not allow the user to view all of the data at once, and it will only allow the user to search the database for a surname.







	2.	Design



The User Interface



The Form for this application requires:

a box for the person’s name

a box for the date of birth

a box for the comment

a button to indicate a new record

a button to activate a search

a button to move to the next record

a button to move to the previous record

a Quit button



A suitable Form might look like this:



�����������	Labels			Text Boxes					Command Buttons

Program Operation





The program operates mainly through the user clicking the different buttons.  However, if the user enters new data into the text boxes, the program needs to store these changes in the file.  Hence some actions will occur as a result of the user typing in the text boxes.  There also needs to be several general procedures: to read and write data to the file, to format data in a uniform manner, and to display the data.  These can go into the module as well.  Lastly, the code will be clearer and easier to read if some of the code for finding a record and adding a new record is also placed in the module as general procedures.





	Event:	The program begins



	Actions:	1.	Create variables to store the name of the data file, a person’s record, the number of records in the file, and the current person’s record number.



		2.	Define the record structure as a data type.



		3.	Set the current record number to 1.



		4.	Open the data file, and read the first person’s record from the file.



		5.	Display the first person’s record.





	Event:	The user clicks the New button



	Actions:	1.	Get the name of the new person.



		2.	Make sure that the name is written in a standard format.



		3.	Insert the new person’s record into the database at the correct location.



		4.	Display the new name on the Form, and set the cursor into the Date of Birth box ready to enter the new birthdate.





	Event:	The user clicks the Find button



	Actions:	1.	Get the surname of the person to find.



		2.	Make sure that this data is in the standard format.



		3.	Find the relevant record in the file.



		4.	Display this record on the Form.





	Event:	The user clicks the Next button



	Actions:	1.	If the current record is not the last record, increase the current record number by 1.



		2.	Read the new current record from the file.



		3.	Display the person’s record.





	Event:	The user clicks the Previous button



	Actions:	1.	If the current record is not the first record, decrease the current record number by 1.



		2.	Read the new current record from the file.



		3.	Display the person’s record.





	Event:	The user clicks the Quit button



	Actions:	1.	Close the data file.



		2.	End the program.





	Event:	The user moves the cursor out of one of the text boxes



	Action:	1.	Save the current record.





It is better to have the record saved when the cursor moves out of the text boxes (the lostFocus event) rather than the change event, because the change event occurs whenever any of the data in the box changes.  Using the change event would have the record saved at every keystroke, and the disk drive would be working continuously.









	3.	Production





Creating The Form And Its Controls





You do not want the user changing the size of the Form in this application.  You can prevent some of the actions which changes a Form’s size by setting its BorderStyle property to 1 - Fixed Single (this prevents the user changing the size of the Form by clicking and dragging), setting the ControlBox property to False (removing this menu and its Size option), and setting the MaxButton property to False (so the Form cannot be maximised).  These settings mean that the Form you create stays the same size when the application is run.  The user can only minimise the Form, or run it at the size you designed.



All of the controls for this Form are ones you have used before.  Make sure that you give them all suitable names.  Do not use the word “name” as the name for a text box.  This word has a special meaning in Visual Basic, and should not be used as a control name.





Entering The Program Code





This program requires the creation of a “user-defined” data type.  This data type will specify the structure of the records in the database.  Visual Basic does not allow you to define a data type at the Form level; it must be defined in the module.  Because this program has to have a module to contain the data type definition, we will place all of the general procedures into the module as well.





Create a new module, and save it as BIRTHDAY.BAS



Into the module, enter this code:

�

Type personrecord

    name As String * 50

    birthdate As String * 20

    comments As String * 200

End Type

Global Const filename = "bday.dat"		Identifies the data file.

Global person As personrecord		Declare three global variables, and their data

Global numpeople As Integer		types

Global currentnum As Integer



Sub initialise ()

�    openfile

    currentnum = 1

    readrecord currentnum

    displayrec person

End Sub



Sub openfile ()

    Open filename For Random Access Read Write As #1 Len = Len(person)

    numpeople = LOF(1) / Len(person)

End Sub



The commands in this last procedure need some detailed explanation.  The Open command opens the file whose name is stored in the filename variable.  Random Access means that the program can go directly to any of the records in the file.  Read Write means that the program can read the data from the file and write new data to the file.  Open files are identified by numbers - this file is number 1.  Len specifies the length of each record in the file; the records are equal to the length of the data stored in the person variable, which is of the personrecord data type.  LOF gives the length of the open file.   



Continue entering these procedures into the module:

�

Sub readrecord (recnum As Integer)

    Get #1, recnum, person

End Sub



Sub writerecord (recnum As Integer)

    Put #1, recnum, person

End Sub



Sub closefile ()

    Close #1

End Sub



Sub displayrec (person As personrecord)

    form1.NameBox.Text = LTrim$(RTrim$(person.name))

    form1.BirthdayBox.Text = LTrim$(RTrim$(person.birthdate))

    form1.CommentBox.Text = LTrim$(RTrim$(person.comments))

End Sub



This procedure also requires some explanation.  The LTrim$ and RTrim$ functions remove any spaces from the left and right end respectively of a string of characters.  The variable named person stores data in the personrecord data type.  Notice how the various parts of that data structure are used to display the record in the boxes on the Form.

The next procedure, which also goes into the module, places a new record in the file, in the correct alphabetical position.  It starts at the end of the file (counter is initially equal to the number of people in the file).  It then gets each record from the file in turn, and compares the name data of the record with the name data in the new record.  If the file record name is greater than (that is, comes later in the alphabet than) the new record name, the file record is placed one position further down in the file, and the next record taken from the file  This process continues until the file record’s name is not greater than the new record’s name, at which point the new record is written into the file.  If no records are found with the name data greater than the name data in the new record, the new record is placed last in the file.



Sub addnewperson (person As personrecord)

    Dim checkitem As personrecord

    counter = numpeople

    found = False

    Do While (found = False) And (counter >= 1)

        Get #1, counter, checkitem

        If RTrim$(checkitem.name) > RTrim$(person.name) Then

            Put #1, counter + 1, person

            counter = counter - 1

        Else

            Put #1, counter + 1, person

            found = True

        End If

    Loop

    If found = False Then

        Put #1, counter + 1, person

    End If

    numpeople = numpeople + 1

    currentnum = counter + 1

End Sub



The next procedure writes a string of data (thename) so that each word starts with a capital letter.  It uses some interesting string functions.  Instr find the first position within a string of another string.  In this procedure, Instr is used to find the first space in the string, and therefore identify the characters in the first word in the string.  Left$ returns a number of characters from the left of a string, Mid$ returns a string of characters starting from some position within the string.  The words in the string are extracted one by one, capitalised using the Ucase function, and then stored in the destination string.



�Sub doformat (thename As String)

    Dim source As String

    Dim Nextword As String

    Dim destination As String

    Dim position As Integer

    destination = ""

    source = LTrim$(RTrim$(thename))

    Do While source <> ""

        position = InStr(1, source, " ")

        If position <> 0 Then

            Nextword = Left$(source, position - 1)

            source = Mid$(source, position)

        Else

            Nextword = source

            source = ""

        End If

        source = LTrim$(RTrim$(source))

        Nextword = UCase$(Left$(Nextword, 1)) + Mid$(Nextword, 2)

        destination = destination + Nextword + " "

    Loop

    thename = LTrim$(RTrim$(destination))

End Sub



The last procedure for the module is used to find a record.  The user supplies the first few characters of the surname, and this procedure finds and displays the first record with name data whose first few characters match the input string.



Sub find (firstpart As String)

    numchars = Len(firstpart)

    counter = 1

    found = False

    Do While (found = False) And (counter < numpeople)

        readrecord counter

        If firstpart = Left$(person.name, numchars) Then

            found = True

        Else

            counter = counter + 1

        End If

    Loop

    If found = True Then

        currentnum = counter

    Else

        Beep

    End If

    readrecord currentnum

    displayrec person

End Sub





The remaining procedures are entered into the appropriate event procedures for the Form:



Sub Form_Load ()

    initialise

End Sub



�Sub BirthdayBox_LostFocus ()

    person.birthdate = birthdaybox.Text

    writerecord currentnum

End Sub



Sub CommentBox_LostFocus ()

    person.comments = commentbox.Text

    writerecord currentnum

End Sub



Sub NameBox_LostFocus ()

    person.name = namebox.Text

    writerecord currentnum

End Sub



Sub NewButton_Click ()

    temp = InputBox$("Enter the person's name (Surname, then First name)", "New Person")

    If temp <> "" Then

�        doformat temp

        person.name = temp

        person.birthdate = ""

        person.comments = ""

        addnewperson person

        displayrec person

        birthdaybox.SetFocus

    End If

End Sub



The remaining procedures should be fairly easy for you to understand by now.  Read them carefully, nevertheless, and make sure you understand how they operate.



Sub FindButton_Click ()

    temp = InputBox$("Enter the first few letters of the surname ...", "Find")

    If temp <> "" Then

        doformat temp

        find temp

    End If

End Sub



Sub NextButton_Click ()

    If currentnum < numpeople Then

        currentnum = currentnum + 1

        readrecord currentnum

        displayrec person

    End If

End Sub



Sub PreviousButton_Click ()

    If currentnum > 1 Then

        currentnum = currentnum - 1

        readrecord currentnum

        displayrec person

    End If

End Sub



Sub QuitButton_Click ()

    closefile

    End

End Sub





This program has introduced quite a lot of fairly complicated code, embodying some difficult and complex programming ideas.  You should, firstly, enter the code and run the program to make sure that it works.  You should then spend some time reading carefully through the different procedures.  They will be difficult for you to understand, and you will be tempted to brush over them and move on to some more interesting programming task.  Nevertheless, your programming abilities will increase significantly if you can grasp the details of the operation of these procedures, and your stock of useful programming techniques will increase as well.

	4.	Evaluation





Testing The Program





1.	The first time you run the program, the data file will not exist.  Add some data (at least three records) to the database; the data file should be saved as you do so. 



	Exit from the program, and use the windows File manager to check that the data file has been created.  The file should be placed in the C:\vb directory.





2.	Run the program again.  Did the database appear?



	Check that the Next and Previous buttons operate as expected.



	Test that the Find function is working properly.





3.	The program should write the data in the Name field with a capital letter first, and the rest lowercase.  Add several new records, using a variety of different capitalisation schemes for the names as you input them.  Try all lower case (alice liddell), all capitals (ALICE LIDDELL), lowercase initial and the rest capitals (aLICE lIDDELL), a mixture (aLIce LIddeLL), and capital initial and the rest lowercase (Alice Liddell).



	Did the data appear as you expected?





4.	The program is supposed to store birthdates.  Enter several of the birthdates in different ways - January 26, 1994 ;  26/1/94 ;  jan 26 94, 26 Jan 94 ; and so on.



How do these dates appear in the text box?  Why?







Improving The Program





1.	Add a text box to the Form to display the record number of the record being displayed.





2.	The program uses the UCase$ function to change the initial letters of the person’s name to capitals.  It does not, however, change any of the other letters in the name.



	Visual Basic has a LCase$ function which will change capital letters to lowercase.  Modify the doformat procedure so that all of the non-initial letters in the names are lowercase.





3.	It would be good if the dates were all written in a uniform format as well.  Look up the Format$ function in the Visual Basic Help, and use it to format the dates in the date boxes into a uniform format.





4.	Use the IsDate function to test the birthdate data as it is input, and present a Message Box to alert the user if a value has been entered that Visual Basic cannot read as a date.





Extension Project:	“The Expanded Birthday Book”





The “Birthday Book” program is not particularly useful as it stands.  It only stores a person’s name and birthdate; it only finds records for a particular surname, and it only arranges the data in alphabetical order of surname.



It would be more useful if the program would do some other things for the user:

calculate, store and display the star sign for each person in the database, based on the birthdate.

calculate, and display and store, the person’s age.

find and display the records with birthdates in a particular month (so that the user can plan presents, cards, etc.).

allow the data to be arranged in a different order (say order of birthdate, or order of age) as well as order of surname.





	Problem Statement:	To develop a database application that allows searching and sorting on several different fields.  Also, to have the program calculate other information from the input data, and display the results of these calculations on the Form.







Adding these features to the database requires some work on your part:



1.	You will, firstly, need to redesign the Form to display additional data fields.  This will also involve changing the structure of the personrecord data type, so that the extra data can be stored in the database file.





2.	You will need to investigate the date functions in Visual Basic, so that your program can extract the month and year of birth from the birthdate.  There is a Now() function which returns the current date, which can be used to calculate the person’s age.





3.	If the data is going to be sorted in some way other than the alphabetical order of surname, the program needs to have all of the data in memory at once.  This will require the use of an array of data of the personrecord data type.  You will need to rewrite the program so that the entire data file is read into memory at the start of the program, and that the entire file is written back to the disk at the end of the program.





4.	None of your programs to date have used a sorting algorithm.  The “Birthday Book” doesn’t sort a set of data, it only inserts a piece of data into its proper place.  You will need to develop an algorithm for rearranging an array of data into some order.  The algorithm will have to use some sort of looping structure, and will be based on comparing two items of data at a time.





5.	The program only displays a single data record at a time.  Add another Form to the application which uses the Grid control to set up a grid of cells on the Form.  Have the program display the data in the database with one record in each row of the grid, so that much more of the data can be seen at the one time.

“The Birthday Book”



These first five lines define a data type called personrecord.  This data type has three parts - name, which is a 50-character string, birthdate, a 20-character string, and comment, a 200-character string.



openfile, displayrec and readrec are all general procedures defined elsewhere in this program.



Go to the open file numbered 1, read the record numbered recnum, and store that record in the person variable.



Declaring these variables within the procedure ensures that the variables, and their values, are known only within this procedure.  They are local variables.



These three procedures are all executed when a LostFocus event occurs.  A text box loses the focus when the cursor is moved out of the box.  When a user enters new data, a text box will get the focus, the user will change the text in the box, and then the box loses the focus.  When it loses the focus, the data in the box is immediately written into the data file.



The InputBox function used in this procedure presents a pre-defined dialog box to the user.  The data which is entered into that box is stored in the temp variable.








